

由应用系统接管 FoxPRO 的错误捕获和处理

罗 辉 (湖南省双峰工商银行)

任何一个系统,在设计和运行过程中总不可避免会隐含各种未可预知的错误,尤其在软件设计日趋庞大的今天,隐含错误的几率越来越大。但只要我们能及时捕获到错误以便纠正,或当错误出现后能妥善处理,那么出现错误并不可怕。对错误的捕获和处理是应用系统设计中挺重要的一部分。但是,许多的应用软件中,错误的捕获和处理并不灵活,能提供的错误信息亦不完善,维护较困难。往往还出现这样的情况:当操作员发现一个错误、维护人员赶到现场时,屏幕上的错误信息却早已消失不见而无法维护!

如果你是一个 XENIX 系统管理员,你将总可以在 XENIX 中发现一些 core 和 errlog 之类的文件,XENIX 系统利用这些文件容纳系统的错误信息。从这些文件中常常可获得一些重要的维护信息。如果我们设计的软件也能提供这样一种错误信息的处理和保存能力,既能反映错误发生的详细情况,并能及时将之保存到一个错误信息文件中,那么将会给软件调试和维护提供很大方便。

本文设计了一个通用错误信息捕获和处理过程 errorproc,如果你是在 FOXPRO 平台上开发软件,那么将之挂靠到你的应用系统中去将是一个很不错的主意。

当应用系统出现错误时,errorproc 过程能自动捕获引起错误的丰富详尽的环境信息,包括出错的文件名、程序行、错误类型、错误提示信息、当前数据库及编辑字段域,以及当前系统状态、内存变量等的取值情况,并保留在一个“黑盒子”文件 ERROR.TXT 中。出错时,你可以现场查阅或打印当前的错误信息,根据出错现场状态即可诊断引起错误的出处和原因。或者,如果出错现场已被破坏,维护人员也可翻阅 ERROR.TXT 文件了解当时的出错情况。有了这样一个错误捕获和处理的自动过程,可弥补操作员没有详细记载现场而难以维护的不

足。尤其在多用户环境下更可见它的优越性。

程序在 FOXPRO2.5 和华达汉字系统 HDDOS1.0 的运行环境下通过。

< 演示调用错误捕获和处理过程的程序 EXAMPLE.PRG >

```
set talk off
clear
on error do errorproc with LINENO(1),program()
select 1
use install
Other Process    &&非法命令,由它激活错误捕获和处理过程
on error
return
```

```
procedure errorproc    &&错误捕获和处理过程
parameter lineno,progfile
private errorno
on error    &&暂时关闭错误陷阱
errorno = error()
do createrrfile    &&形成错误信息文件
DEFINE WINDOW error ;
FROM INT((SROW()-15)/2),INT((SCOL()-50)/2);
TO INT((SROW()-15)/2)+16,INT((SCOL()-50)/2)+54;
TITLE " 错误信息窗口 " SHADOW DOUBLE;
COLOR SCHEME 10
```

ACTIVATE WINDOW error

DO WHILE .T.

```
@ 3,2 GET choice ;
PICTURE "@ * H 重试;退出;挂起;忽略;显示;打印;清除";
SIZE 1,6,1 DEFAULT 1
@ 7,2 SAY "出错号:"+ltrim(str(errorno))
@ 7,20 SAY "出错行号:"+ltrim(str(lineno))
@ 8,2 SAY "出错程序名:"+progfile
@ 9,2 SAY "出错信息:"
@ 10,2 SAY message()
```

```

READ CYCLE
on error do errorproc with LINENO(1),PROGRAM()
    &&恢复错误陷阱

DO CASE
CASE CHOICE=1
    RELEASE WINDOW error
    RETRY
CASE CHOICE=2
    RELEASE WINDOW error
    QUIT
CASE CHOICE=3
    DEACTIVATE WINDOW error
    SUSPEND
    ACTIVATE WINDOW error
CASE CHOICE=4
    RELEASE WINDOW error
    EXIT
CASE CHOICE=5
    modi comm tmp.txt noedit
CASE CHOICE=6
    @12,2 GET dychoice pict "@ * H 打印当前信息;打印历史信息";
    size 1,16,1 default 1
    read
    dyfile=if(dychoice#2,'tmp.txt','error.txt')
    if FILE((DYFILE))
        ! type &dyfile. > PRN    && DOS 外部命令不支持
(dyfile)等名字表达式
    ENDIF
    @12,2 clear to 12,30
CASE CHOICE=7
    IF FILE('ERROR.TXT')
        ! DEL ERROR.TXT > NUL
    ENDIF
ENDCASE
ENDDO
RETURN

procedure createrrfile    &&错误信息文件形成过程
    set console off
    set printer on
    set printer to tmp.txt
    ? padc("错误汇报",78,'*')
    ? 'Date:'+dtoc(date())+space(5)+'Time:'+time()
    ?? space(5)+'FoxPro 版本号:'

```

```

?? version()
?
proccount = 1
do while len(sys(16,proccount))<>0 &&嵌套调用的程序名
    ? if(proccount = 1,'主程序:',space(proccount-1)+'调用:')
    ?? sys(16,proccount)
    proccount = proccount+1
enddo
?
? '错误号:'+ltrim(str(errorno))
? '错误信息:'+message()
? '引起错误的程序行号:'+str(lineno)
? '出错行语句:'
? message(1)
?
IF len(alltrim(dbf()))#0
    ? '当前工作区:'
    ?? select()
    ? '打开的数据库:'+DBF()
    ndxnum = 1
    SCAN WHILE LEN(NDX(NDXNUM))#0
        ? NDXNUM
        ?? '索引文件名:'+NDX(NDXNUM)
        NDXNUM = NDXNUM+1
    ENDSCAN
ENDIF
tmp=ALLTRIM(VARREAD())
if LEN(tmp)<>0
    ? '当前的编辑域:'+tmp
    ? '其取值为:'
    ?? (tmp)
ENDIF
?
? '出错时的状态:'
list stat
?
? '出错时的内存变量或数组取值:'
list memo
? repl(' ',80)
! copy error.txt+tmp.txt error.txt > NUL
set printer to
set printer off
set console on
return

```